G05EGF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G05EGF sets up a reference vector for an autoregressive moving-average (ARMA) time series model with Normally distributed errors, so that G05EWF may be used to generate successive terms. It also initialises the series to a stationary position.

2 Specification

```
SUBROUTINE G05EGF(E, A, NA, B, NB, R, NR, VAR, IFAIL)
INTEGER           NA, NB, NR, IFAIL
real              E, A(*), B(NB), R(NR), VAR
```

3 Description

The ARMA model of such a time series in discrete time is

$$(x_n - E) = A_1(x_{n-1} - E) + \ldots + A_{NA}(x_{n-NA} - E) + B_1\epsilon_n + \ldots + B_{NB}\epsilon_{n-NB+1}$$

where x_n is the value of the series at time n, and ϵ_n is a series of independent random Standard Normal perturbations.

The routine copies A, E and B to the reference vector so that G05EWF can generate the terms of the series. It sets up initial values corresponding to a stationary position using the method described in Tunnicliffe–Wilson [2].

4 References

[1] Knuth D E (1981) *The Art of Computer Programming (Volume 2)* Addison–Wesley (2nd Edition)

[2] Tunnicliffe–Wilson G (1979) Some efficient computational procedures for high order ARMA models *J. Statist. Comput. Simulation* **8** 301–309

5 Parameters

1: E — ***real*** *Input*

On entry: the mean of the time series.

2: A(∗) — ***real*** array *Input*

Note: the dimension of the array A must be at least $\max(1, \mathrm{NA})$.

On entry: the autoregressive coefficients of the model, if any.

3: NA — INTEGER *Input*

On entry: the number of autoregressive coefficients supplied.

Constraint: $\mathrm{NA} \geq 0$.

4: B(NB) — ***real*** array *Input*

On entry: the moving-average coefficients of the model.

5: NB — INTEGER *Input*

On entry: the number of moving-average coefficients supplied.

Constraint: $\mathrm{NB} \geq 1$.

6: R(NR) — ***real*** array *Output*

On exit: the reference vector and the recent history of the series.

7: NR — INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05EGF is called.

Constraint: NR $\geq$ NA + NB + 4 + max(NA, NB).

8: VAR — ***real*** *Output*

On exit: the proportion of the variance of a term in the series that is due to the moving-average (error) terms in the model. The smaller this is, the nearer is the model to non-stationarity.

9: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, NA < 0.

IFAIL = 2

On entry, NB < 1.

IFAIL = 3

On entry, NR $<$ NA + NB + 4 + max(NA, NB).

IFAIL = 4

A does not define a stationary autoregressive process.

7 Accuracy

The errors in the initialisation process should be very much smaller than the error term; see Tunnicliffe–Wilson [2].

8 Further Comments

The time taken by the routine is essentially of order $(\text{NA})^2$.

Note. G05CBF, G05CCF, G05CFF, and G05CGF must be used with care if this routine is used as well. The reference vector, as mentioned before, contains a copy of the recent history of the series. This will not be altered properly by calls to any of the above routines. A call to G05CBF or G05CCF should be followed by calls to G05EGF to re-initialise all time series reference vectors in use. To maintain repeatability with G05CBF, the calls to G05EGF should be performed in the same order and at the same point or points in the simulation every time G05CBF is used. When routines G05CFF and G05CGF are used to save and restore the generator state, all the time series reference vectors in use must be saved and restored as well.

The ARMA model for a time series can also be written as:

$$(x_t - c) = \phi_1(x_{t-1} - c) + \ldots + \phi_p(x_{t-p} - c) + a_t - \theta_1 a_{t-1} \ldots - \theta_q a_{t-q}$$

where x_t is the observed value of the time series at time t,

p is the number of autoregressive parameters, ϕ_i,

q is the number of moving average parameters, θ_i,

c is the mean of the time series,

and a_t is a series of independent random Normal perturbations with variance σ^2.

This is the form used in the G13 Chapter Introduction. This is related to the form given in Section 3 by:

$B_1^2 = \sigma^2$,

$B_{i+1} = -\theta_i\sigma = -\theta_i B_1, \quad i = 1, 2,,$

$NB = q + 1$,

$E = c$,

$A_i = \phi_i, \quad i = 1, 2, \ldots, p.$

and NA $= p$.

9 Example

This example program calls G05EGF to set up the reference vector for the autoregressive model

$$x_n = 0.4x_{n-1} + 0.2x_{n-2} + \epsilon_n$$

where ϵ_n is a series of independent random Standard Normal perturbations. G05EWF is then called 10 times to generate a sample of observations, which are printed.

The generator mechanism used is selected by an initial call to G05ZAF.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05EGF Example Program Text
*     NAG Fortran SMP Library, Release 2.  NAG Copyright 2000.
*     .. Parameters ..
      INTEGER          NA, NB, NR
      PARAMETER        (NA=2,NB=1,NR=NA+NB+4+NA)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION VAR, X
      INTEGER          I, IFAIL
*     .. Local Arrays ..
      DOUBLE PRECISION A(NA), B(NB), R(NR)
*     .. External Functions ..
      DOUBLE PRECISION G05EWF
      EXTERNAL         G05EWF
*     .. External Subroutines ..
      EXTERNAL         G05CBF, G05EGF, G05ZAF
*     .. Executable Statements ..
      CALL G05ZAF('O')
      WRITE (NOUT,*) 'G05EGF Example Program Results'
      WRITE (NOUT,*)
      CALL G05CBF(0)
      A(1) = 0.4D0
      A(2) = 0.2D0
      B(1) = 1.0D0
```

```
      IFAIL = 0
*
      CALL G05EGF(0.0D0,A,NA,B,NB,R,NR,VAR,IFAIL)
*
      DO 20 I = 1, 10
         IFAIL = 0
         X = G05EWF(R,NR,IFAIL)
         WRITE (NOUT,99999) X
   20 CONTINUE
      STOP
*
99999 FORMAT (1X,F12.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05EGF Example Program Results

       2.4084
       1.1987
       2.4778
       0.7998
       0.0452
       0.4125
       0.3784
      -1.2166
      -0.3510
       1.1631
```